```
> restart:
```

Ranada.mws (R. M. Kiehn, Sept 9 1997)

In "Essays on Classical and Quantum Dynamics" , edited by J.A. Ellison and H. Uberall (Gordon Breach) A. F. Ranada gave an example of an electromagnetic field constructed from the Hopf map from S3 to S2. The field had a finite helicity integral, which is a signature of a longitudinal B field. The field satisfied the Poincare (projective) condition of a characteristic, (E dot B)=0, and therefore was a candidate or a wave packet solution in electromagnetism..

The process can be generalized by using a projective representation (renormalization) of the four vector of potentials. The standard projection (renormalization) process for an arbitrary vector field is to define a characteristic function, Ch, of the components of the vector field, and divide the original vector by this function. When the denominator function Ch is linear, the zero set of this function defines the "point at infinity" in classical projective geometry. When the function is (homogeneously) quadratic in the components of the vector field, the zero set defines a projective conic (Example, the light cone).

The process can be generalized to an arbitrary homogeneous function, Ch.

Applied to an arbitrary vector field with components [U,V,W,T], the projective renormalization (Characteristic denominator) function is defined as

```
> Ch:=(a*U^p+b*V^p+c*W^p+e*T^p)^(n/p);
```

$$Ch := (a\,U^p + b\,V^p + c\,W^p + e\,T^p)^{\left(\frac{n}{p}\right)}$$

```
>
```

This format is to recognized as the Holder norm, and is a function homogenous of degree n. The anisotropy coefficients (a,b,c,e) are arbitrary. Such projective vectors, treated as position vectors, have a Jacobian matrix with zero trace for n=1, zero determinant when n = 4, and represents a reflexive (involution) when m=2. By dividing through by say eT^p the stereographic projective format - used by Ranada - can be obtained,

```
>  Chstereo:=(a*(U/T)^p+b*(V/T)^p+c*(W/T)^p +1)^(n/p);
```

$$Chstereo := \left(a\left(\frac{U}{T}\right)^p + b\left(\frac{V}{T}\right)^p + c\left(\frac{W}{T}\right)^p + 1\right)^{\left(\frac{n}{p}\right)}$$

The Ranada example is just one of many many possibilities. Another possibility built upon a Holder type norm demonstrates that B3 vanishes if the homogeneity is of degree n=2. (The involution case). But for all other n or p, B3 is not zero in the class of examples shown.

****

NONE of this implies that the constitutive constraints are those with the symmetries of the vacuum.

NONE of this implies some symmetry group is dominant.

****

Setup and initialize for the Holder type solution:

```
>  with(liesymm):with(linalg):with(plots):setup(x,y,z,t);
Warning, new definition for close
Warning, new definition for norm
Warning, new definition for trace
```

$$[x, y, z, t]$$

```
> defform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,omega=const,m=const);
```

$$\mathrm{defform}(x = 0, y = 0, z = 0, t = 0, a = const, b = const, c = const, k = const, \mu = const, \omega = const, m = const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [d(x), d(y), d(z), d(t)]$$

Define a characteristic denominator, and then specify the four functions that are the covariant components of the Action 1-form.

```
> rrxy:=a*x^p+b*y^p+c*z^p+e*t^p:ff:=2/(Pi*(rrxy)^(n/p));
> c:=0;e:=0;
```

$$ff := \frac{2}{\pi\,(a\,x^p + b\,y^p + c\,z^p + e\,t^p)^{\left(\frac{n}{p}\right)}}$$

$$c := 0$$

$$e := 0$$

A choice for the 4 potentials.

```
> A1:=y*ff;A2:=-x*ff;A3:=k*ff;A4:=-omega*ff;
```

$$A1 := 2\,\frac{y}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}$$

$$A2 := -2\,\frac{x}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}$$

$$A3 := 2\,\frac{k}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}$$

$$A4 := -2\,\frac{\omega}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}$$

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t);
```

$$Action := 2\,\frac{y\,d(x)}{\pi\,\%1} - 2\,\frac{x\,d(y)}{\pi\,\%1} + 2\,\frac{k\,d(z)}{\pi\,\%1} + 2\,\frac{\omega\,d(t)}{\pi\,\%1}$$

$$\%1 := (a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}$$

```
> F:=wcollect(d(Action));
```

$$F := \left(2\,\frac{-a\,x^p - b\,y^p + n\,b\,y^p}{\pi\,\%1\,(a\,x^p + b\,y^p)} - 2\,\frac{a\,x^p + b\,y^p - n\,a\,x^p}{\pi\,\%1\,(a\,x^p + b\,y^p)}\right)((d(x))\;\&\wedge\;(d(y))) - 2\,\frac{\omega\,n\,a\,x^p\,((d(x))\;\&\wedge\;(d(t)))}{\pi\,\%1\,x\,(a\,x^p + b\,y^p)}$$
$$- 2\,\frac{\omega\,n\,b\,y^p\,((d(y))\;\&\wedge\;(d(t)))}{\pi\,\%1\,y\,(a\,x^p + b\,y^p)} - 2\,\frac{k\,n\,a\,x^p\,((d(x))\;\&\wedge\;(d(z)))}{\pi\,\%1\,x\,(a\,x^p + b\,y^p)} - 2\,\frac{k\,n\,b\,y^p\,((d(y))\;\&\wedge\;(d(z)))}{\pi\,\%1\,y\,(a\,x^p + b\,y^p)}$$

$$\%1 := (a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

The three components of the Vector potential are:

```
> A:=[A1,A2,A3];
```

$$A := \left[2\,\frac{y}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}},\; -2\,\frac{x}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}},\; 2\,\frac{k}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}\right]$$

The three components of the Magnetic field are:

```
> B:=(curl(A,[x,y,z])):B1:=factor(B[1]);B2:=factor(B[2]);B3:=factor(B[3]);
```

$$B1 := -2 \frac{k\,n\,b\,y^p}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\,y\,(a\,x^p + b\,y^p)}$$

$$B2 := 2 \frac{k\,n\,a\,x^p}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\,x\,(a\,x^p + b\,y^p)}$$

$$B3 := 2 \frac{-2 + n}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}}$$

Note that B3 vanishes for n=2 !!!!!

The three components of the Electric Field are

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
  E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);
```

$$E1 := -2 \frac{\omega\,n\,a\,x^p}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\,x\,(a\,x^p + b\,y^p)}$$

$$E2 := -2 \frac{\omega\,n\,b\,y^p}{\pi\,(a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\,y\,(a\,x^p + b\,y^p)}$$

$$E3 := 0$$

Topological Parity 4 form (Second Poncare invariant)

```
> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E):BdotB:=innerprod(B,B):ExB:=c
  rossprod(E,B):
```

$$EdotB := 0$$

The Torsion current.

```
> ExA:=crossprod(E,A);Bphi:=[B1*A4,B2*A4,B3*A4];
```

$$ExA := \left[-4 \frac{\omega\,n\,b\,y^p\,k}{\pi^2\,\%1^2\,y\,(a\,x^p + b\,y^p)}, 4 \frac{\omega\,n\,a\,x^p\,k}{\pi^2\,\%1^2\,x\,(a\,x^p + b\,y^p)}, 4 \frac{\omega\,n\,a\,x^p}{\pi^2\,\%1^2\,(a\,x^p + b\,y^p)} + 4 \frac{\omega\,n\,b\,y^p}{\pi^2\,\%1^2\,(a\,x^p + b\,y^p)}\right]$$

$$\%1 := (a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}$$

$$Bphi := \left[4 \frac{\omega\,n\,b\,y^p\,k}{\pi^2 \left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2 y\,(a\,x^p + b\,y^p)}, -4 \frac{\omega\,n\,a\,x^p\,k}{\pi^2 \left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2 x\,(a\,x^p + b\,y^p)}, -4 \frac{(-2 + n)\,\omega}{\pi^2 \left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2}\right]$$

```
> TORS:=evalm(ExA+A4*B);
```

$$TORS := \left[0, 0, 4 \frac{\omega\,n\,a\,x^p}{\pi^2\,\%1^2\,(a\,x^p + b\,y^p)} + 4 \frac{\omega\,n\,b\,y^p}{\pi^2\,\%1^2\,(a\,x^p + b\,y^p)} - 2 \frac{\omega \left(-\frac{4}{\pi\,\%1} + 2 \frac{n\,a\,x^p}{\pi\,\%1\,(a\,x^p + b\,y^p)} + 2 \frac{n\,b\,y^p}{\pi\,\%1\,(a\,x^p + b\,y^p)}\right)}{\pi\,\%1} \right]$$

$$\%1 := (a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}$$

```
> AdotB:=factor(inner(A,B));
```

$$AdotB := -8 \frac{k}{\left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2 \pi^2}$$

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$$TORSION := \left[0,\,0,\,8 \frac{\omega}{\left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2 \pi^2},\,-8 \frac{k}{\left((a\,x^p + b\,y^p)^{\left(\frac{n}{p}\right)}\right)^2 \pi^2}\right]$$

Divergence of the Torsion current.

```
> DIVT:=factor(diverge(TORSION,[x,y,z,t]));
```

$$DIVT := 0$$

Just to show how the symbolic mechanism in Maple works, the following initializes the code for the a Ranada type solution:

```
>  restart:with(liesymm):with(linalg):with(plots):setup(x,y,z,t);
Warning, new definition for close
Warning, new definition for norm
Warning, new definition for trace
```

$$[x, y, z, t]$$

```
> defform(x=0,y=0,z=0,t=0,a=const,b=const,c=const,k=const,mu=const,omega=const,m=c
  onst);
```

$$\mathrm{defform}(x = 0,\, y = 0,\, z = 0,\, t = 0,\, a = const,\, b = const,\, c = const,\, k = const,\, \mu = const,\, \omega = const,\, m = const)$$

```
> dR:=[d(x),d(y),d(z),d(t)];
```

$$dR := [\mathrm{d}(x), \mathrm{d}(y), \mathrm{d}(z), \mathrm{d}(t)]$$

Define a characteristic denominator, and then specify the four functions that are the covariant components of the Action 1-form. Note that a stereographic projection is used.

```
> rrxy:=1+a*x^p+b*y^p+c*z^p:ff:=2/(Pi*(rrxy)^(n/p));
```

$$ff := \frac{2}{\pi\,(1 + a\,x^p + b\,y^p + c\,z^p)^{\left(\frac{n}{p}\right)}}$$

```
> n:=4;p:=2;a:=1;b:=1;c:=1;
```

$$n := 4$$

$$p := 2$$

$$a := 1$$

$$b := 1$$

$$c := 1$$

A possible choice for the 4 potentials. Note that the Ranada example presumes A4 = constant. This assumption eliminates the E field!. Note that n=4, and p =2 in the Ranada case. You can modify the coefficients and get different results using Maple.

```
> A1:=y*ff;A2:=-x*ff;A3:=k*ff;A4:=omega;
```

$$A1 := 2 \frac{y}{\pi\,(1 + x^2 + y^2 + z^2)^2}$$

$$A2 := -2 \frac{x}{\pi\,(1 + x^2 + y^2 + z^2)^2}$$

$$A3 := 2 \frac{k}{\pi\,(1 + x^2 + y^2 + z^2)^2}$$

$$A4 := \omega$$

$$ff := \frac{2}{\pi\,(a\,x^p + b\,y^p + c\,z^p + e\,t^p)^{\left(\frac{n}{p}\right)}}$$

$$c := 0$$

$$e := 0$$

```
> Action:=A1*d(x)+A2*d(y)+A3*d(z)-A4*d(t);
```

$$Action := 2\,\frac{y\,\mathrm{d}(x)}{\pi\,(1+x^2+y^2+z^2)^2} - 2\,\frac{x\,\mathrm{d}(y)}{\pi\,(1+x^2+y^2+z^2)^2} + 2\,\frac{k\,\mathrm{d}(z)}{\pi\,(1+x^2+y^2+z^2)^2} - \omega\,\mathrm{d}(t)$$

```
> F:=wcollect(d(Action));
```

$$F := \left(-8\,\frac{k\,x}{\pi\,(1+x^2+y^2+z^2)^3} + 8\,\frac{y\,z}{\pi\,(1+x^2+y^2+z^2)^3}\right)((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(z))) + \left(-2\,\frac{1+x^2-3\,y^2+z^2}{\pi\,(1+x^2+y^2+z^2)^3} - 2\,\frac{1-3\,x^2+y^2+z^2}{\pi\,(1+x^2+y^2+z^2)^3}\right)((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(y))) + \left(-8\,\frac{k\,y}{\pi\,(1+x^2+y^2+z^2)^3} - 8\,\frac{x\,z}{\pi\,(1+x^2+y^2+z^2)^3}\right)((\mathrm{d}(y))\ \&\wedge\ (\mathrm{d}(z)))$$

F is the electromagnetic 2-form in covariant language for all diffeomorphisms. It is gauge invariant with respect to all closed 1-form additions to the 1-form of Action.

The three components of the Vector potential are:

```
> A:=[A1,A2,A3];
```

$$A := \left[2\,\frac{y}{\pi\,(1+x^2+y^2+z^2)^2},\ -2\,\frac{x}{\pi\,(1+x^2+y^2+z^2)^2},\ 2\,\frac{k}{\pi\,(1+x^2+y^2+z^2)^2}\right]$$

The three components of the Magnetic field are:

```
> B:=(curl(A,[x,y,z])):B1:=factor(B[1]);B2:=factor(B[2]);B3:=factor(B[3]);
```

$$B1 := -8\,\frac{k\,y + x\,z}{\pi\,(1+x^2+y^2+z^2)^3}$$

$$B2 := -8\,\frac{y\,z - k\,x}{\pi\,(1+x^2+y^2+z^2)^3}$$

$$B3 := -4\,\frac{1 - x^2 - y^2 + z^2}{\pi\,(1+x^2+y^2+z^2)^3}$$

Note that B3 does not vanish

The three components of the Electric Field are zero by default definition of the potentials

```
> E:=[-diff(A4,x)-diff(A[1],t),-diff(A4,y)-diff(A[2],t),-diff(A4,z)-diff(A[3],t)]:
  E1:=factor(E[1]);E2:=factor(E[2]);E3:=factor(E[3]);
```

$$E1 := 0$$

$$E2 := 0$$

$$E3 := 0$$

Topological Parity 4 form (Second Poncare invariant)

```
> EdotB:=factor(innerprod(E,B));EdotE:=innerprod(E,E):BdotB:=innerprod(B,B):ExB:=c
  rossprod(E,B):
```

$$EdotB := 0$$

The Torsion current.

```
> ExA:=crossprod(E,A);Bphi:=[B1*A4,B2*A4,B3*A4];
```

$$ExA := [0, 0, 0]$$

$$Bphi := \left[-8\frac{(k\,y+x\,z)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3}, -8\frac{(y\,z-k\,x)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3}, -4\frac{(1-x^2-y^2+z^2)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3} \right]$$

```
> TORS:=evalm(ExA+A4*B);
```

$$TORS := \left[\omega\left(-8\frac{k\,y}{\pi\,(1+x^2+y^2+z^2)^3} - 8\frac{x\,z}{\pi\,(1+x^2+y^2+z^2)^3}\right), \omega\left(-8\frac{y\,z}{\pi\,(1+x^2+y^2+z^2)^3} + 8\frac{k\,x}{\pi\,(1+x^2+y^2+z^2)^3}\right), \omega\left(-\frac{4}{\pi\,(1+x^2+y^2+z^2)^2} + 8\frac{x^2}{\pi\,(1+x^2+y^2+z^2)^3} + 8\frac{y^2}{\pi\,(1+x^2+y^2+z^2)^3}\right) \right]$$

```
> AdotB:=factor(inner(A,B));
```

$$AdotB := -8\frac{k}{(1+x^2+y^2+z^2)^4\,\pi^2}$$

```
> TORSION:=[factor(TORS[1]),factor(TORS[2]),factor(TORS[3]),AdotB];
```

$$TORSION := \left[-8\frac{(k\,y+x\,z)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3}, -8\frac{(y\,z-k\,x)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3}, -4\frac{(1-x^2-y^2+z^2)\,\omega}{\pi\,(1+x^2+y^2+z^2)^3}, -8\frac{k}{(1+x^2+y^2+z^2)^4\,\pi^2} \right]$$

Divergence of the Torsion current.

```
> DIVT:=factor(diverge(TORSION,[x,y,z,t]));
```

$$DIVT := 0$$

Ranada also assumed that k=1 and omega= 0. The latter constraint got rid of the spatial part of the Topological Torsion vector.

```
>
>
```